Scaled Bernoulli numbers

This worksheet is a supplement to: Marc Mezzarobba, *Rounding Error Analysis of Linear Recurrences Using Generating Series*. It illustrates how some of the calculations of the section on scaled Bernoulli numbers can be checked using Maple.

```
> restart;
```

Define the auxiliary series S-caron, b-sharp-star. Double-check the associated inequalities.

```
> Sc := w/sin(w);
```

$$Sc := \frac{w}{\sin(w)} \tag{1}$$

```
> series(w/sinh(w), w, 12);
```

$$1 - \frac{1}{6} w^2 + \frac{7}{360} w^4 - \frac{31}{15120} w^6 + \frac{127}{604800} w^8 - \frac{73}{3421440} w^{10} + O(w^{12}) \tag{2}$$

```
> series(Sc, w, 12);
```

$$1 + \frac{1}{6} w^2 + \frac{7}{360} w^4 + \frac{31}{15120} w^6 + \frac{127}{604800} w^8 + \frac{73}{3421440} w^{10} + O(w^{12}) \tag{3}$$

```
> bs := 2 - w/tan(w);
```

$$bs := 2 - \frac{w}{\tan(w)} \tag{4}$$

```
> seq(bernoulli(2*k)/(2*k)!, k=0..5);
```

$$1, \frac{1}{12}, -\frac{1}{720}, \frac{1}{30240}, -\frac{1}{1209600}, \frac{1}{47900160} \tag{5}$$

```
> series(subs(w=sqrt(z)/2, bs), z);
```

$$1 + \frac{1}{12} z + \frac{1}{720} z^2 + \frac{1}{30240} z^3 + \frac{1}{1209600} z^4 + \frac{1}{47900160} z^5 + O(z^6) \tag{6}$$

Define C-tilde-star, S-tilde-star.

```
> a := 1 + u;
```

$$a := 1 + u \tag{7}$$

```
> Ct := cosh(a*w) - cosh(w);
```

$$Ct := \cosh((1+u)\, w) - \cosh(w) \tag{8}$$

```
> St := w^(-1)*(sinh(a^2*w) - sinh(w)) - (a^2-1);
```

$$St := \frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} - (1+u)^2 + 1 \tag{9}$$

Now define maj = delta-hat-star in terms of the series introduced so far.

```
> num := Ct+St*bs
```

$$num := \left(\frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} - (1+u)^2 + 1\right)\left(2 - \frac{w}{\tan(w)}\right) + \cosh((1+u)\, w) - \cosh(w) \tag{10}$$

```
> den := 1/Sc-St
```

$$den := \frac{\sin(w)}{w} - \frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} + (1+u)^2 - 1 \tag{11}$$

```
> maj := num/den
```

$$maj := \left(\left(\frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} - (1+u)^2 + 1\right)\left(2 - \frac{w}{\tan(w)}\right) + \cosh((1+u)\, w) - \cosh(w)\right) \Big/ \left(\frac{\sin(w)}{w} - \frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} + (1+u)^2 - 1\right) \tag{12}$$

(We could have used solve(), but the form above will be more convenient to work with than the one solve() returns.)

```
> normal(maj - solve(maj1 = Sc*Ct + Sc*St*bs + Sc*St*maj1, maj1));
```

(13)

Behavior of maj as u → 0. First, the series expansion with respect to u.

```
> ser := simplify(convert(series(maj, u, 2), exp), exp);
```

$$ser := \frac{\left(\left(\left(e^{2w}+1\right)e^{-w}-2\right)\left(2-\frac{w\cos(w)}{\sin(w)}\right)+\frac{\left(e^{2w}-1\right)e^{-w}w}{2}\right)w}{\sin(w)}\,u+O\left(u^2\right) \tag{14}$$

Extract the leading term xi-hat and rewrite it in a nicer form.

```
> xih := op(1, ser);
```

$$xih := \frac{\left(\left(\left(e^{2w}+1\right)e^{-w}-2\right)\left(2-\frac{w\cos(w)}{\sin(w)}\right)+\frac{\left(e^{2w}-1\right)e^{-w}w}{2}\right)w}{\sin(w)} \tag{15}$$

```
> xih := collect(expand(xih), [sin, cos, w], e -> simplify(convert(e, trig), trig));
```

$$xih := \frac{\sinh(w)\,w^2+(4\cosh(w)-4)\,w}{\sin(w)}+\frac{(-2\cosh(w)+2)\,w^2\cos(w)}{\sin(w)^2} \tag{16}$$

Singular expansion of x-hat at its poles. We isolate the singular factor for readability.

```
> aux := xih*(1 - (w/Pi)^2)^2;
```

$$aux := \left(\frac{\sinh(w)\,w^2+(4\cosh(w)-4)\,w}{\sin(w)}+\frac{(-2\cosh(w)+2)\,w^2\cos(w)}{\sin(w)^2}\right)\left(1-\frac{w^2}{\pi^2}\right)^2 \tag{17}$$

```
> simplify(MultiSeries[series](aux, w=Pi, 4));
```

$$8\cosh(\pi)-8+\frac{4\pi\sinh(\pi)+8\cosh(\pi)-8}{\pi}\,(w-\pi)+O\left((w-\pi)^2\right) \tag{18}$$

The function studied in Lemma 18, i.e., the denominator of maj divided by Sc.

```
> h := St - 1/Sc;
```

$$h := \frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} - (1+u)^2 + 1 - \frac{\sin(w)}{w} \quad (19)$$

```
> eval(h, u=0)
```

$$-\frac{\sin(w)}{w} \quad (20)$$

```
> plot(eval(h, u=1e-10), w=-3*Pi..3*Pi)
```



The term St vanishes identically when u=0, leaving us with the zeros of sin(w)/w. We focus on the zero at w=π. The Implicit Function Theorem applies, so that the zero varies analytically with u for small enough u.

```
> eval(diff(h, w),[w=Pi,u=0]);
```

$$\frac{1}{\pi} \quad (21)$$

```
> phi := RootOf(subs(w=Pi/(1+_Z), h)):
> phiser := simplify(convert(series(phi, u, 3), trig));
```

$$phiser := \left(2\cosh(\pi) - 2\right) u + \left(-2\pi\sinh(2\pi) + 6\pi\sinh(\pi) + \cosh(\pi) - 1\right) u^2 + O\left(u^3\right) \quad (22)$$

```
> theK := op(1,phiser)
```

$$theK := 2\cosh(\pi) - 2 \quad (23)$$

```
> G := sinh(w) - w - w^3/6
```

$$G := \sinh(w) - w - \frac{w^3}{6} \quad (24)$$

Check that $a^2 w0 < \pi$

```
> w0 := Pi/(1+K*u)
```

(25)

$$w0 := \frac{\pi}{K\,u+1} \qquad (25)$$

```
> plot(subs(K=theK, a^2*w0/Pi), u=0..1/2)
```




```
> solve(diff(a^2*w0, u), u)
```

$$-1,\ \frac{K-2}{K} \qquad (26)$$

Double-check some algebraic identities used in the proof

```
> normal((w0-Pi) + K*u*w0)
```

$$0 \qquad (27)$$

```
> St_alt := (a^6-1)*w^2/6 + (subs(w=a^2*w, G) - G)/w
```

$$St\_alt := \frac{\left((1+u)^6-1\right)w^2}{6} + \frac{\sinh\left((1+u)^2 w\right) - (1+u)^2 w - \frac{(1+u)^6 w^3}{6} - \sinh(w) + w + \frac{w^3}{6}}{w} \qquad (28)$$

```
> normal(St-St_alt)
```

$$0 \qquad (29)$$

Upper bound on h(w0), and a quick check that it is negative

```
> hhat := (a^6-1)/6*w0^2 + (a^2-1)*(K-Pi^2)/(2) - K*u + (K*u)^3*w0^2/6
```

$$hhat := \frac{\left((1+u)^6-1\right)\pi^2}{6\,(K\,u+1)^2} + \frac{\left((1+u)^2-1\right)\left(-\pi^2+K\right)}{2} - K\,u + \frac{K^3 u^3 \pi^2}{6\,(K\,u+1)^2} \qquad (30)$$

```
> sort(collect(numer(hhat)/6, [u,Pi]), u, ascending)
```

(31)

$$\left((-2K+2)\pi^2+\frac{K}{2}\right)u^2+\left(\left(\frac{1}{6}K^3-K^2-K+\frac{10}{3}\right)\pi^2+K^2\right)u^3+\left(\left(-\frac{K^2}{2}+\frac{5}{2}\right)\pi^2+\frac{K^3}{2}\right)u^4+\pi^2u^5+\frac{\pi^2u^6}{6} \quad (31)$$

```
> plot(subs(K = op(1,phiser), hhat), u=-2^(-8)..2^(-5))
```




```
> RootFinding[Analytic](diff(subs(K = op(1,phiser), numer(hhat)), u), u, re=0..2^(-5), im=-1..1)
```

$$0.0223706820632626, 0.\,I \quad (32)$$

|sin(w)|, |w|=ρ is minimal for real w. Argument vaguely inspired by https://math.stackexchange.com/a/1322980: write w=ρ(x+iy). The expression cosh(ρ·y) reaches its global minimum at y=0, corresponding to x=±1, and is increasing for y≥0. To the left of x=1, cos(ρ·x) is initially decreasing up to x=3π/(2ρ), where cosh(y) is already large enough that $\cosh(\rho y)^2-\cos(\rho x)^2 \geq \cosh(\rho y)^2-1 > |\sin(C)|^2$.

```
> sqabssin := simplify(abs(sin(rho*exp(I*theta)))^2) assuming theta::real
```

$$sqabssin := \left|\sin\left(\rho\, e^{I\theta}\right)\right|^2 \quad (33)$$

```
> rho := 62/10
```

$$\rho := \frac{31}{5} \quad (34)$$

```
> plot(sqabssin, theta=-Pi..Pi)
```



Check the condition of Rouché's theorem. Maple's support for rigorous numerical computations is poor; we redo the check using interval arithmetic in the accompanying SageMath worksheet.

```
> evalf(eval(expand(a^2-1)*(1+cosh(a^2*rho)), u=2^(-16)))
```

0.007550782584 **(35)**

```
> evalf(abs(sin(rho)/rho))
```

0.01340151659 **(36)**

Singular expansion at alpha of delta-hat-star (proof of Proposition 19)

```
> zz_alpha := RootOf(-h, w); # apparently it needs to be -h, not h, for MultiSeries[series] to succeed
```

$$zz\_alpha := RootOf\left(\_Z\,u^2 + 2\,u\,\_Z - \sinh\left((1+u)^2\,\_Z\right) + \sinh(\_Z) + \sin(\_Z)\right) \tag{37}$$

```
> alias(alpha = zz_alpha)
```

$$\alpha \tag{38}$$

```
> ser := MultiSeries[series](maj, w=alpha, 2)
```

$$ser := \Big(\big(-\cos(\alpha)\,\alpha^2 u^2 - 2\cos(\alpha)\,\alpha^2 u + 2\sin(\alpha)\,\alpha\,u^2 + \cosh(\alpha)\sin(\alpha)\,\alpha + \cos(\alpha)\,\alpha\sinh\left(\alpha\,(1+u)^2\right) - \cos(\alpha)\,\alpha\sinh(\alpha) - \cosh(\alpha\,(1+u))\sin(\alpha)\,\alpha + 4\sin(\alpha)\,\alpha\,u - 2\sin(\alpha)\sinh\left(\alpha\,(1+u)^2\right) + 2\sin(\alpha)\sinh(\alpha)\big)\,\alpha\Big) \Big/ \Big(\sin(\alpha)\,\big(\cosh\left(\alpha\,(1+u)^2\right)\alpha\,u^2 + 2\cosh\left(\alpha\,(1+u)^2\right)\alpha\,u + \cosh\left(\alpha\,(1+u)^2\right)\alpha - \alpha\cosh(\alpha) - \alpha\cos(\alpha) + \sin(\alpha) - \sinh\left(\alpha\,(1+u)^2\right) + \sinh(\alpha)\big)\Big)\,(-\alpha + w)^{-1} + \mathrm{O}\left((-\alpha + w)^0\right) \tag{39}$$

```
> rew := isolate(subs(_Z=alpha,op(zz_alpha)),sinh((1+u)^2*alpha))
```

$$rew := \sinh\left(\alpha\,(1+u)^2\right) = \alpha\,u^2 + 2\,\alpha\,u + \sinh(\alpha) + \sin(\alpha) \tag{40}$$

```
> ser := simplify(subs(rew,ser))
```

$$ser := \frac{-\alpha\cosh(\alpha\,(1+u)) + (\cos(\alpha) + \cosh(\alpha))\,\alpha - 2\sin(\alpha)}{(1+u)^2\cosh\left(\alpha\,(1+u)^2\right) - u^2 - 2\,u - \cos(\alpha) - \cosh(\alpha)}\,(-\alpha + w)^{-1} + \mathrm{O}\left((-\alpha + w)^0\right) \tag{41}$$

The auxiliary function R

```
> R := normal(convert(ser, polynom)*(1-w/alpha)):
> R := subs(_a=alpha,simplify(combine(subs(alpha=_a, R))))
```

$$R := \frac{-\alpha\cosh(\alpha) - \alpha\cos(\alpha) + \alpha\cosh(\alpha\,(1+u)) + 2\sin(\alpha)}{\left((1+u)^2\cosh\left(\alpha\,(1+u)^2\right) - u^2 - 2\,u - \cos(\alpha) - \cosh(\alpha)\right)\alpha} \tag{42}$$

Double-check the expression for R given in the paper

```
> simplify(1/R*(subs(w=alpha,Ct)-cos(alpha)+2*sin(alpha)/alpha)/(subs(w=alpha,diff(w*St,w))-cos(alpha)))
```

$$1 \tag{43}$$

Asymptotics of R as u → 0

```
> serR := simplify(subs(_Z1=1,MultiSeries:-series(R, u, 2)))
```

$$serR := 1 + (-1)^{-\_Z1}\left(\pi\,\_Z1\sinh(\pi\,\_Z1) - 2\cosh(\pi\,\_Z1) + 2\right)u + \mathrm{O}\left(u^2\right) \tag{44}$$

```
> evalf(serR)
```

$$1. + (-1.)^{-\_Z1}\,(3.141592654\,\_Z1\sinh(3.141592654\,\_Z1) - 2.\cosh(3.141592654\,\_Z1) + 2.)\,u + \mathrm{O}\left(u^2\right) \tag{45}$$

The derivative of R. After a non-rigorous look at its values, we generate a straight-line program implementing the expression that we use in the accompanying SageMath notebook to compute rigorous enclosures.

```
> dR := (diff(R, u)):
> plot(subs(alpha=Pi/(1+theK*u), dR), u=0..2^(-16))
```




```
> lprint(codegen[optimize](dR))
t1 = alpha, t3 = 1+u, t4 = t1*t3, t5 = t3^2, t6 = t5*t1, t7 = cosh
(t6), t9 = -
t1*u+t4*t7-t1, t10 = u^2, t11 = 2*u, t13 = cosh(t1), t14 = cos(t1),
t15 = -t5*
t7+t10+t11+t13+t14, t16 = 1/t15, t17 = 2*t16*t9, t19 = 2*t9*t1, t20 =
sinh(t1),
t24 = sin(t1), t27 = cosh(t4), t32 = sinh(t4), t35 = -1/t15, t37 =
1/t1, t43 =
-t1*t13-t1*t14+t1*t27+2*t24, t44 = t15^2, t53 = sinh(t6), t61 = t1^2,
t66 = t37
*t35*(-t13*t17-t20*t16*t19+t14*t17+t24*t16*t19+t27*t17+t32*(t17*t3+
t1)*t1)-(2*
t7*t3+t53*(t17*t5+2*t4)*t5-t11-2+t24*t17-t20*t17)*t37/t44*t43-2*t16*
t9/t61*t35*
t43
```

Similar steps with g to estimate A[λ].

```
> g := maj - 2*R/(1-(w/alpha)^2) - 2/(1-(w/Pi)^2)
```

$$g := \left(\left(\frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} - (1+u)^2 + 1 \right) \left(2 - \frac{w}{\tan(w)} \right) + \cosh((1+u)\, w) - \cosh(w) \right) \Bigg/ \left(\frac{\sin(w)}{w} - \frac{\sinh\left((1+u)^2 w\right) - \sinh(w)}{w} + (1+u)^2 - 1 \right) - \frac{2\,(-\alpha \cosh(\alpha) - \alpha \cos(\alpha) + \alpha \cosh(\alpha\,(1+u)) + 2 \sin(\alpha))}{\left((1+u)^2 \cosh\left(\alpha\,(1+u)^2\right) - u^2 - 2\,u - \cos(\alpha) - \cosh(\alpha)\right) \alpha \left(1 - \frac{w^2}{\alpha^2}\right)} \tag{46}$$

$$-\frac{2}{1-\frac{w^2}{\pi^2}}$$

```
> lambda := sqrt(3/2):
> dg := diff(g, u):
> _dg := subs({alpha=Pi/(1+theK*u)}, dg):
> plot(theta->abs(evalf(eval(_dg, [u=evalf(2^(-16)), w=evalf(lambda*Pi*exp(I*theta))]
  ))), -Pi..Pi)
```




```
> lprint(codegen[optimize](dg))
t1 = 1+u, t2 = t1^2, t3 = w*t2, t4 = cosh(t3), t6 = t1*t4-u-1, t7 =
tan(w), t10
= 2-1/t7*w, t12 = w*t1, t13 = sinh(t12), t16 = 1/w, t17 = sin(w), t19
= sinh(t3
), t20 = sinh(w), t22 = (t19-t20)*t16, t23 = t16*t17+t2-t22-1, t28 =
cosh(t12),
t29 = cosh(w), t31 = t23^2, t35 = alpha, t37 = t35*t1, t38 = t2*t35,
t39 = cosh
(t38), t41 = -t35*u+t37*t39-t35, t42 = u^2, t43 = 2*u, t45 = cosh
(t35), t46 =
cos(t35), t47 = -t2*t39+t42+t43+t45+t46, t48 = 1/t47, t49 = 2*t48*
t41, t51 = 2*
t41*t35, t52 = sinh(t35), t56 = sin(t35), t59 = cosh(t37), t64 = sinh
(t37), t67
= -1/t47, t70 = w^2, t71 = t35^2, t72 = 1/t71, t74 = -t70*t72+1, t75
= 1/t74,
t76 = t75/t35, t83 = -t35*t45-t35*t46+t35*t59+2*t56, t84 = t47^2, t93
= sinh(
t38), t101 = t67*t83, t107 = t71^2, t110 = t74^2, t116 = 1/t23*(2*
t10*t6+t13*w)
```

```
+2*t6/t31*(t10*(t22-t2+1)+t28-t29)-2*t76*t67*(-t45*t49-t52*t48*t51+
t46*t49+t56*
t48*t51+t59*t49+t64*(t1*t49+t35)*t35)+2*(2*t39*t1+t93*(t2*t49+2*t37)*
t2-t43-2+
t56*t49-t52*t49)*t76/t84*t83+4*t48*t41*t75*t72*t101+4*t49*
t70/t110/t107*t101
```